

Konačne grupe

Sage putem sučelja prema GAP sustavu za računalnu algebru ima implementiran niz operacija s konačnim i Lievim grupama.

```
C2 = CyclicPermutationGroup(2)
C3 = CyclicPermutationGroup(3)
D3 = DihedralGroup(3)
```

```
print C3
D3
```

```
Cyclic group of order 3 as a permutation group
Dihedral group of order 6 as a permutation group
```

Broj elemenata grupe:

```
print C3.order()
D3.order()
```

```
3
6
```

```
print C3.is_abelian()
D3.is_abelian()
```

```
True
False
```

Popis svih elemenata grupe kao permutacija. Za notaciju vidi [ovdje](#).

```
print C3.list()
D3.list()

[(), (1,2,3), (1,3,2)]
[(), (2,3), (1,2), (1,2,3), (1,3,2), (1,3)]
```

Rad s pojedinim elementima grupe:

```
c = D3("(1,2,3)")
b = D3("(2,3)")
b2 = D3("(1,3)")
```

Treba uočiti da se kompozicija u GAP-u/Sage-u izvrijednjuje s lijeva na desno, a ne obratno kao na predavanjima!

```
c^-1*b*c == b2 # relacija s predavanja
```

```
True
```

```
(b*c)^2 # definiciona relacija za dihedralne grupe
```

```
()
```

Provjera ove iste relacije za grupu D_7 :

```
D7 = DihedralGroup(7); D7
```

```
Dihedral group of order 14 as a permutation group
```

```
cc = D7('(1,2,3,4,5,6,7)')
```

```
bb = D7('(2,7)(3,6)(4,5)')
```

```
(bb*cc)^2 == D7('()')
```

```
True
```

Reprezentanti klasa konjugacije:

```
print C3.conjugacy_classes_representatives()
D3.conjugacy_classes_representatives()
```

```
[(), (1,2,3), (1,3,2)]
```

```
[(), (1,2), (1,2,3)]
```

... i cijele klase:

```
for el in D3.conjugacy_classes_representatives():
    print "\nclass of %s:" % str(el)
```

```
    print list(set([g^-1*el*g for g in D3.list()]))
```

```
class of ():
```

```
[()]
```

```
class of (1,2):
```

```
[(2,3), (1,2), (1,3)]
```

```
class of (1,2,3):
```

```
[(1,2,3), (1,3,2)]
```

Grupna tablica množenja:

```
TC3 = C3.cayley_table(); TC3
```

$$\begin{array}{r}
 * \quad a \quad b \quad c \\
 +----- \\
 a| \quad a \quad b \quad c \\
 b| \quad b \quad c \quad a \\
 c| \quad c \quad a \quad b
 \end{array}$$

Legenda:

```
C3.cayley_table().row_keys()
(), (1,2,3), (1,3,2))
```

To odgovara elementima $\{e, c, c^2\}$ iz predavanja.

```
TC3.change_names(['e', 'c', 'c^2']); TC3
```

$$\begin{array}{r}
 * \quad e \quad c \quad c^2 \\
 +----- \\
 e| \quad e \quad c \quad c^2 \\
 c| \quad c \quad c^2 \quad e \\
 c^2| \quad c^2 \quad e \quad c
 \end{array}$$

```
latex.eval(TC3._latex_(), {}, "")
```

$$\begin{array}{c|ccc}
 \times & e & c & c^2 \\
 \hline
 e & e & c & c^2 \\
 c & c & c^2 & e \\
 c^2 & c^2 & e & c
 \end{array}$$

Skup svih podgrupa grupe D_3 :

```
D3.subgroups()
```

```
[Permutation Group with generators [()],
 Permutation Group with generators [(2,3)],
 Permutation Group with generators [(1,2)],
 Permutation Group with generators [(1,3)],
 Permutation Group with generators [(1,2,3)],
 Permutation Group with generators [(1,3,2),
 (1,3)]]
```

To su trivijalna podgrupa ($\{e\}$), zatim tri C_2 podgrupe, C_3 podgrupa i sama grupa D_3

```
D3.subgroups()[-2] == C3
```

True

Lijeve susjedne klase podgrupe C_3 u D_3 :

```
D3.cosets(C3, side='left')
[[(), (1,2,3), (1,3,2)], [(2,3), (1,2), (1,3)]]
```

Kvocijentni skup D_3/C_3 je jednak grupi C_2 :

```
print D3.quotient(C3) == C2
print C3.is_normal(D3) # C3 je normalna podgrupa od D3 ...
C2.is_normal(D3)       #... ali C2 nije
True
True
False
```

Tablica karaktera:

```
D3.character_table()
```

```
[ 1 -1  1]
[ 2  0 -1]
[ 1  1  1]
```

Ljepši ispis može se dobiti izravnom komunikacijom sa GAP paketom:

```
print gap.eval("Display(%s)" % gap(D3).CharacterTable().name())
```

CT2

```
2  1  1  .
3  1  .  1

      1a 2a 3a
2P 1a 1a 3a
3P 1a 2a 1a

X.1    1 -1  1
X.2    2  . -1
X.3    1  1  1
```

Treba uočiti da je i redoslijed klasa i redoslijed reprezentacija u tablici drugačiji nego na predavanjima.
Redoslijed klase odgovara redoslijedu u listi koju daje metoda
`conjugacy_classes_representatives()`

```
C3.character_table()
```

```
[           1           1           1]
[           1       zeta3 -zeta3 - 1]
[           1 -zeta3 - 1       zeta3]
```

```
print gap.eval("Display(%s)" % gap(C3).CharacterTable().name())
```

CT1

3 1 1 1

1a 3a 3b

X.1	1	1	1
X.2	1	A	/A
X.3	1	/A	A

$$\begin{aligned} A &= E(3) \\ &= (-1 + \sqrt{-3})/2 = b3 \end{aligned}$$

Ovdje je $\zeta_3 = b3 = E(3) = e^{2\pi i/3} = (-1 + i\sqrt{3})/2$

Literatura: Robert A. Beezer, [Group Theory and Sage](#).